```
!
! Magnolia CSL model file created on 2019-04-10T15:55:16.793
! PBPK Gas Uptake Model for 1,1-DCPe converted from original acslx model by EMK
! Units of this model are liters (L), hours (h), milligrams (mg), BW in kilograms (kg)

model DCPe11

initial  ! The INITIAL section contains statements evaluated 1x @ begin simulation

!***********! Physiological Parameters**********************}
CONSTANT    QPC = 13.2                          ! Alveolar Ventilation Rate(l/hr-kg), Clewell et al., 2014, Table 6.1, p. 264
CONSTANT    QCC = 13.2                          ! Cardiac Output(l/hr-kg), Clewell et al., 2014, Table 6.1, p. 264
CONSTANT   QLC = 0.174                          ! Fract. Blood Flow to Liver, Table 25, p. 439, Brown et al., 1997
CONSTANT   QFC = 0.070                          ! Fract. Blood Flow to Fat, Table 25, p. 439, Brown et al., 1997
CONSTANT   QSC = 0.336                          ! Fract. Blood Flow to Poorly Perfused T., , Table 25, p. 439, Brown et al., 1997 (muscle + skin)
        QRC = 1.0 - QLC - QFC - QSC         ! Fract. Blood Flow to Richly Perfused T

CONSTANT   BW = 0.222                           ! Body Weight(kg), experiment-specific, mean & SD all rats 0.222+/-0.0184
CONSTANT   VLC = 0.0349                         ! Volume Fraction Liver, Brown et al., 1997, Table 8, p. 418
CONSTANT   VFC = 0.0798                         ! Volume Fraction Fat, Brown et al., 1997, equation, p. 422 using avg BW
CONSTANT   VSC = 0.5946                         ! Volume Fraction Poorly Perfused T., muscle + skin, Table 5, p. 416, Brown et al., 1997
       VRC = 1.0 - VLC - VFC - VSC  ! Volume Fraction Richly Perfused T.

 !***********! Chemical Specific Parameters for DCPs**********************}
 ! Partition Coeffs Estimated Based on method of Beliveau et al., (2003)}
CONSTANT      PL =  1.71      ! Liver/Blood
CONSTANT      PF  = 34.85     ! Fat/Blood
CONSTANT      PS  = 0.17      ! Poorly/Blood (muscle is representative tissue)
CONSTANT      PR  = 1.71      ! Richly/Blood (liver is representative tissue)
CONSTANT      PB  = 5.97      ! Blood/Air
CONSTANT      MW  = 110.97    ! Molecular Wt (g/mol)

!VmaxC & Km starting values - final values are estimated from set of vapor uptake curves
CONSTANT     VmaxC = 5.58   ! mg/hr-kg
```

```
CONSTANT       Km = 0.23   ! mg/L

 ! Chamber/Rat/Exposure Parameters}
CONSTANT      CONC = 500.                          ! Init. Inhaled Conc (ppm), experiment-specific
CONSTANT      RATS = 1                             ! Number of rats in chamber, always 1
CONSTANT      VCHC = 3.8                           ! Closed chamber vol (l)
CONSTANT      KLOS = 0.04                          ! Closed chamber loss, hr-1, experiment-specific
    NETVOL = VCHC - (RATS * BW)                    ! Net Chamber Vol (l), i.e. volume of chamber not taken up by rats
      AIO = (CONC * NETVOL * MW)/24450             ! Initial Conc of chemical in Chamber (mg/l)
CONSTANT   TSTOP = 6.00                            ! Length of experiment (hrs)
CONSTANT  points = 100.                            ! No. of points in plot

!*******Scaled Parameters********}

       QC = QCC * (BW**0.75)     ! Cardiac Output (l/hr)
       QP = QPC * (BW**0.75)     ! Alveolar Ventilation Rate (l/hr)
       QL = QLC * QC             ! Flow Liver Compartment (l/hr)
       QF = QFC * QC             ! Flow Fat Compartment (l/hr)
       QS = QSC * QC             ! Flow Slowly Perf. Tis. Cmpt. (l/hr)
       QR = QRC * QC             ! Flow Richly Perf. Tis. Cmpt. (l/hr)
       VL = VLC * BW             ! Volume Liver Compartment, Total
       VF = VFC * BW             ! Volume Fat Compartment
       VS = VSC * BW             ! Volume Slowly Perfused Tis. Cmpt.
       VR = VRC * BW             ! Volume Richly Perfused Tis. Cmpt.

  VMAX = VMAXC * BW**0.75    ! VMAX scaled

end ! initial

dynamic   !  The DYNAMIC section statements evaluated at each output time point

   derivative

      RAI = (RATS * QP * (CX - CI)) - (KLOS*AI)      ! mg/hr
```

```
      AI = INTEG(RAI, AIO)                  ! mg
      CI = AI / NETVOL                                    ! mg/l
      CP = (CI* 24450/MW)                                 ! chamber conc, ppm, data is in ppm

! chamber loss
    RLOSS = (KLOS * AI)                   ! mg/hr
     LOSS = INTEG(RLOSS, 0.0)

  CA = ((QC * CV) + (QP * CI)) / (QC + (QP / PB))  ! arterial (mg/L)

 !exhaled breath
      RAX = QP * CX                       ! mg/hr
       AX = INTEG(RAX, 0.0)                  ! mg
       CX = CA/PB                                      ! Conc. DCPe in exhaled air(mg/l)
    CXPPM = ((0.7 * CX) + (0.3 * CI)) * (24.45/MW)        ! Conc. DCPE in ppm  exhaled breath
      AXKG = AX/BW                                        ! mg exhaled/kg body weight

!slowly perfused tissue group
   RAS = QS * (CA - CVS)            ! mg/hr
    AS = INTEG(RAS,0.0)             ! mg
   CVS = AS / (VS * PS)                          ! Conc partition to slow per. tis.(mgl)
   CS  = AS / VS                                 ! Conc in volume slow per. tis.(mg/l)

!rapidly perfused tissue group
   RAR = QR * (CA - CVR)             ! mg/hr
    AR = INTEG(RAR, 0.0)             ! mg
   CVR = AR / (VR * PR)                          ! Conc partition to rap per. tis.(mg/l)
    CR = AR / VR                        ! Conc in volume rap per. tis.(mg/l)

!fat
   RAF = QF * (CA - CVF)            ! mg/hr
    AF = INTEG(RAF, 0.0)            ! mg
   CVF = AF / (VF * PF)                          ! Conc partition to fat(mg/l)
   CF  = AF / VF                                 ! Conc in fat volume(mg/l)
```

```
!liver
   RAL = QL * (CA - CVL) - RAM        ! mg/hr
    AL = INTEG(RAL, 0.0)             ! mg
   CVL = AL / (VL * PL)                              ! Conc partition to liver(mg/l)
    CL = AL / VL                                    ! Conc in liver volume(mg/l)

   RAM = (VMAX * CVL) / (KM + CVL)
    AM = INTEG(RAM, 0.0)

   CV = (QF*CVF + QL*CVL + QS*CVS + QR*CVR) / QC       ! venous blood conc (mg/l)

   DOSE = AIO - AI                                                    ! amt dcp in rat (mg)

   DMASS = (AF + AL + AS + AR + AM)*RATS

!Criterion - mass balance (MB)  should be near zero, i.e. <= ~e10-8.}

   MB = DOSE - DMASS - LOSS          ! mass balance (mg)

   TERMT(T >= TSTOP)                 ! Condition for terminating simulation

end ! derivative

end ! dynamic

end ! program
```

```
! Parameter estimation script for 1,1-DCPe PBPK model
! Closed-chambed rat data

load '11DCPe.csl'

data @file='dcp1_50a.csv' ds1  t='t' cp='cp' tstop=2.5 bw=0.205 klos=0.025 conc=49
data @file='dcp1_50b.csv' ds2  t='t' cp='cp' tstop=3.33 bw=0.213 klos=0.039 conc=48
data @file='dcp1_50c.csv' ds3  t='t' cp='cp' tstop=3.33 bw=0.201 klos=0.049 conc=44
data @file='dcp1_120.csv' ds4  t='t' cp='cp' tstop=2.5 bw=0.243 klos=0.028 conc=93
data @file='dcp1_200a.csv' ds5  t='t' cp='cp' tstop=4.83 bw=0.219 klos=0.0604 conc=181
data @file='dcp1_200b.csv' ds6  t='t' cp='cp' tstop=4.83 bw=0.222 klos=0.063 conc=185
data @file='dcp1_200c.csv' ds7  t='t' cp='cp' tstop=5 bw=0.213 klos=0.048 conc=173
data @file='dcp1_500a.csv' ds8  t='t' cp='cp' tstop=5 bw=0.224 klos=0.0785 conc=429
data @file='dcp1_500b.csv' ds9  t='t' cp='cp' tstop=5.17 bw=0.212 klos=0.014 conc=409
data @file='dcp1_500c.csv' ds10  t='t' cp='cp' tstop=5.17 bw=0.219 klos=0.017 conc=417
data @file='dcp1_1000a.csv' ds11 t='t' cp='cp' tstop=6 bw=0.225 klos=0.067 conc=901
data @file='dcp1_1000b.csv' ds12 t='t' cp='cp' tstop=6 bw=0.229 klos=0.0446 conc=885
data @file='dcp1_1000c.csv' ds13 t='t' cp='cp' tstop=6 bw=0.251 klos=0.058 conc=952
data @file='dcp1_1000d.csv' ds14 t='t' cp='cp' tstop=5.83 bw=0.213 klos=0.0407 conc=940

! A procedure to plot all data sets
! Used to plot the fits before and after parameter estimation
procedure plotall

   set tstop=2.5 bw=0.205 klos=0.025 conc=49
   start
   plot cp 'ds1:cp' @title='50a'

   set tstop=3.33 bw=0.213 klos=0.039 conc=48
   start
   plot cp 'ds2:cp' @title='50b'

   set tstop=3.33 bw=0.201 klos=0.049 conc=44
   start
```

```
plot cp 'ds3:cp' @title='50c'

set tstop=2.5 bw=0.243 klos=0.028 conc=93
start
plot cp 'ds4:cp' @title='120'

set tstop=4.83 bw=0.219 klos=0.0604 conc=181
start
plot cp 'ds5:cp' @title='200a'

set tstop=4.83 bw=0.222 klos=0.063 conc=185
start
plot cp 'ds6:cp' @title='200b'

set tstop=5 bw=0.213 klos=0.048 conc=173
start
plot cp 'ds7:cp' @title='200c'

set tstop=5 bw=0.224 klos=0.0785 conc=429
start
plot cp 'ds8:cp' @title='500a'

set tstop=5.17 bw=0.212 klos=0.014 conc=409
start
plot cp 'ds9:cp' @title='500b'

set tstop=5.17 bw=0.219 klos=0.017 conc=417
start
plot cp 'ds10:cp' @title='500c'

set tstop=6 bw=0.225 klos=0.067 conc=901
start
plot cp 'ds11:cp' @title='1000a'
```

```
   set tstop=6 bw=0.229 klos=0.0446 conc=885
   start
   plot cp 'ds12:cp' @title='1000b'

   set tstop=6 bw=0.251 klos=0.058 conc=952
   start
   plot cp 'ds13:cp' @title='1000c'

   set tstop=5.83 bw=0.213 klos=0.0407 conc=940
   start
   plot cp 'ds14:cp' @title='1000d'

end


prepare @clear T CP

set vmaxc = 5.58 !@min=1 @max=10
set km    = 0.28 @min=0.1 @max=2.0

plotall

! Can comment in or out any of the fit statements to run particular scenario
!fit @method=ml @errormodel=mixed vmaxc km
fit @method=ml @errormodel=mixed vmaxc
!fit @method=ml @errormodel=mixed km

plotall
```

```
! Magnolia CMD script created on 2020-02-24T16:12:57.165
! Filename:  SAg_11DCPe.cmd
! Sensitivity Analysis 11-DCPe Global Morris Method

load '11DCPe.csl'

set    QPC = 13.2  @min=10 @max=16    ! Alveolar Ventilation Rate(l/hr-kg), Clewell et al., 2014, Table 6.1, p. 264
set    QCC = 13.2  @min=10 @max=16    ! Cardiac Output(l/hr-kg), Clewell et al., 2014, Table 6.1, p. 264
set    QLC = 0.242 @min=0.2 @max=0.28          !Fract. Blood Flow to Liver, Table 25, p. 439, Brown et al., 1997
set    QFC = 0.082 @min=0.05 @max=0.11 ! Fract. Blood Flow to Fat, Table 25, p. 439, Brown et al., 1997
set    QSC = 0.257 @min=0.23 @max=0.28         ! Fract. Blood Flow to Poorly Perfused T., , Table 25, p. 439, Brown et al., 1997 (muscle + skin)

set   BW = 0.222                        ! Body Weight(kg), experiment-specific, mean & SD all rats 0.222+/-0.0184
set   VLC = 0.036   @min=0.028 @max=0.044 ! Volume Fraction Liver, Brown et al., 1997, Table 8, p. 418
set   VFC = 0.076           @min=0.05 @max=0.09             ! Volume Fraction Fat, Brown et al., 1997, equation, p. 422 using avg BW
set   VSC = 0.7             @min=0.65 @max=0.75   ! Volume Fraction Poorly Perfused T., muscle + skin, Table 5, p. 416, Brown et al., 1997

  !************! Chemical Specific Parameters for DCPs************************}
  ! Partition Coeffs Estimated Based on method of Beliveau et al., (2003)}
  ! Range set assuming 30% CV and 1 SD
set       PL =  1.71    @min=1.2 @max=2.2   ! Liver/Blood
set       PF  = 34.85   @min=10.4 @max=45.3   ! Fat/Blood
set       PS  = 0.17    @min=0.12 @max=0.22   ! Poorly/Blood (muscle is representative tissue)
set       PR  = 1.71    @min=1.2 @max=2.2   ! Richly/Blood (liver is representative tissue)
set       PB  = 5.97    @min=4.2 @max=7.8   ! Blood/Air
set       MW  = 110.97     ! Molecular Wt (g/mol)

!VmaxC & Km starting values - final values are estimated from set of vapor uptake curves
!Using same range setting as in Parameter Estimation simulations
set      VmaxC = 5.58   @min=1.0 @max=10     ! mg/hr-kg
set      Km = 0.23      @min=0.1 @max=2.0    ! mg/L


! Chamber/Rat/Exposure Parameters}
```

```
set      CONC = 1000.      ! Init. Inhaled Conc (ppm), experiment-specific, use 50, 200, 500 and 1000 ppm
set      RATS = 1            ! Number of rats in chamber, always 1
set      VCHC = 3.8          ! Closed chamber vol (l), constant
set      KLOS = 0.04         @min=0.02 @max=0.06          ! Closed chamber loss, hr-1, experiment-specific, range used actual
set     TSTOP = 6.00      ! Length of exposure

prepare t, cp, mb

output @clear t cp mb

start @nocallback
plot @title = 'Gas Uptake GSA Setup 11-DCPe', @xlabel = 'Time (hrs)', @ylabel = 'CP (ppm)', cp

! Global sensitivity analysis, Morris method, response is cp, setting reflect same values used in JAT Ms.
gsa @method=morris @jump= 25 @nlevels=100 @nsamps=1000 @response=cp @coeffile='cp_gsa_11DCPe.csv' QPC QCC QLC QFC QSC VLC VFC
VSC...
PL PF PS PR PB VMAXC KM KLOS
line @xvar=t @file='cp_gsa_11DCPe.csv' QPC QCC QLC QFC QSC VLC VFC VSC...
PL PF PS PR PB VMAXC KM KLOS
```